DESCRIPTION

Catalyst is a modern framework for making web applications without the pain usually associated with this process. This document is a reference to the main Catalyst application. If you are a new user, we suggest you start with Catalyst::Manual::Tutorial or Catalyst::Manual::Intro.

Catalyst plugins can be loaded by naming them as arguments to the "use Catalyst" statement. Omit the Catalyst::Plugin:: prefix from the plugin name, i.e., Catalyst::Plugin::My::Module becomes My::Module.

use Catalyst qw/My::Module/;

If your plugin starts with a name other than Catalyst::Plugin::, you can fully qualify the name by using a unary plus:

use Catalyst qw/
My::Module
+Fully::Qualified::Plugin::Name
/;

Special flags like -Debug and -Engine can also be specified as arguments when Catalyst is loaded:

use Catalyst qw/-Debug My::Module/;

The position of plugins and flags in the chain is important, because they are loaded in the order in which they appear.

The following flags are supported:

-Debug

Enables debug output. You can also force this setting from the system environment with CATALYST_DEBUG or <MYAPP>_DEBUG. The environment settings override the application, with <MYAPP>_DEBUG having the highest priority.

This sets the log level to 'debug' and enables full debug output on the error screen. If you only want the latter, see $c->debug.

-Engine

Forces Catalyst to use a specific engine. Omit the Catalyst::Engine:: prefix of the engine name, i.e.:

use Catalyst qw/-Engine=CGI/;

-Home

Forces Catalyst to use a specific home directory, e.g.:

use Catalyst qw[-Home=/usr/mst];

This can also be done in the shell environment by setting either the CATALYST_HOME environment variable or MYAPP_HOME; where MYAPP is replaced with the uppercased name of your application, any "::" in the name will be replaced with underscores, e.g. MyApp::Web should use MYAPP_WEB_HOME. If both variables are set, the MYAPP_HOME one will be used.

If none of these are set, Catalyst will attempt to automatically detect the home directory. If you are working in a development envirnoment, Catalyst will try and find the directory containing either Makefile.PL, Build.PL or dist.ini. If the application has been installed into the system (i.e. you have done make install), then Catalyst will use the path to your application module, without the .pm extension (ie, /foo/MyApp if your application was installed at /foo/MyApp.pm)

-Log

use Catalyst '-Log=warn,fatal,error';

Specifies a comma-delimited list of log levels.

-Stats

Enables statistics collection and reporting.

use Catalyst qw/-Stats=1/;

You can also force this setting from the system environment with CATALYST_STATS or <MYAPP>_STATS. The environment settings override the application, with <MYAPP>_STATS having the highest priority.

METHODS

INFORMATION ABOUT THE CURRENT REQUEST

$c->action

$c->namespace

Returns the namespace of the current action, i.e., the URI prefix corresponding to the controller of the current action. For example:

# in Controller::Foo::Bar
$c->namespace; # returns 'foo/bar';

$c->request

$c->req

Returns the current Catalyst::Request object, giving access to information about the current client request (including parameters, cookies, HTTP headers, etc.). See Catalyst::Request.

REQUEST FLOW HANDLING

$c->forward( $action [, \@arguments ] )

$c->forward( $class, $method, [, \@arguments ] )

Forwards processing to another action, by its private name. If you give a class name but no method, process() is called. You may also optionally pass arguments in an arrayref. The action will receive the arguments in @_ and $c->req->args. Upon returning from the function, $c->req->args will be restored to the previous values.

Any data returned from the action forwarded to, will be returned by the call to forward.

Note that forward implies an eval { } around the call (actually execute does), thus de-fatalizing all 'dies' within the called action. If you want die to propagate you need to do something like:

$c->forward('foo');
die join "\n", @{ $c->error } if @{ $c->error };

Or make sure to always return true values from your actions and write your code like this:

$c->forward('foo') || return;

Another note is that $c->forward always returns a scalar because it actually returns $c->state which operates in a scalar context. Thus, something like:

return @array;

in an action that is forwarded to is going to return a scalar, i.e. how many items are in that array, which is probably not what you want. If you need to return an array then return a reference to it, or stash it like so:

$c->stash->{array} = \@array;

and access it from the stash.

$c->detach( $action [, \@arguments ] )

$c->detach( $class, $method, [, \@arguments ] )

$c->detach()

The same as forward, but doesn't return to the previous action when processing is finished.

When called with no arguments it escapes the processing chain entirely.

$c->visit( $action [, \@captures, \@arguments ] )

$c->visit( $class, $method, [, \@captures, \@arguments ] )

Almost the same as forward, but does a full dispatch, instead of just calling the new $action / $class->$method. This means that begin, auto and the method you go to are called, just like a new request.

In addition both $c->action and $c->namespace are localized. This means, for example, that $c->action methods such as name, class and reverse return information for the visited action when they are invoked within the visited action. This is different from the behavior of forward, which continues to use the $c->action object from the caller action even when invoked from the callee.

$c->stash is kept unchanged.

In effect, visit allows you to "wrap" another action, just as it would have been called by dispatching from a URL, while the analogous go allows you to transfer control to another action as if it had been reached directly from a URL.

$c->go( $action [, \@captures, \@arguments ] )

$c->go( $class, $method, [, \@captures, \@arguments ] )

The relationship between go and visit is the same as the relationship between forward and detach. Like $c->visit, $c->go will perform a full dispatch on the specified action or method, with localized $c->action and $c->namespace. Like detach, go escapes the processing of the current request chain on completion, and does not return to its caller.

$c->response

$c->res

$c->stash

Returns a hashref to the stash, which may be used to store data and pass it between components during a request. You can also set hash keys by passing arguments. The stash is automatically sent to the view. The stash is cleared at the end of a request; it cannot be used for persistent storage (for this you must use a session; see Catalyst::Plugin::Session for a complete system integrated with Catalyst).

$c->error

$c->error($error, ...)

$c->error($arrayref)

Returns an arrayref containing error messages. If Catalyst encounters an error while processing a request, it stores the error in $c->error. This method should only be used to store fatal error messages.

my @error = @{ $c->error };

Add a new error.

$c->error('Something bad happened');

$c->state

Contains the return value of the last executed action. Note that << $c->state >> operates in a scalar context which means that all values it returns are scalar.

$c->clear_errors

Clear errors. You probably don't want to clear the errors unless you are implementing a custom error screen.

$c->model($name)

If the name is omitted, it will look for - a model object in $c->stash->{current_model_instance}, then - a model name in $c->stash->{current_model}, then - a config setting 'default_model', or - check if there is only one model, and return it if that's the case.

$c->view($name)

If the name is omitted, it will look for - a view object in $c->stash->{current_view_instance}, then - a view name in $c->stash->{current_view}, then - a config setting 'default_view', or - check if there is only one view, and return it if that's the case.

Cascading configuration

The config method is present on all Catalyst components, and configuration will be merged when an application is started. Configuration loaded with Catalyst::Plugin::ConfigLoader takes precedence over other configuration, followed by configuration in your top level MyApp class. These two configurations are merged, and then configuration data whose hash key matches a component name is merged with configuration for that component.

The configuration for a component is then passed to the new method when a component is constructed.

$c->log

Returns the logging object instance. Unless it is already set, Catalyst sets this up with a Catalyst::Log object. To use your own log class, set the logger with the __PACKAGE__->log method prior to calling __PACKAGE__->setup.

__PACKAGE__->log( MyLogger->new );
__PACKAGE__->setup;

And later:

$c->log->info( 'Now logging with my own logger!' );

Your log class should implement the methods described in Catalyst::Log.

$c->debug

Returns 1 if debug mode is enabled, 0 otherwise.

You can enable debug mode in several ways:

By calling myapp_server.pl with the -d flag

With the environment variables MYAPP_DEBUG, or CATALYST_DEBUG

The -Debug option in your MyApp.pm

By declaring sub debug { 1 } in your MyApp.pm.

The first three also set the log level to 'debug'.

Calling $c->debug(1) has no effect.

$c->dispatcher

$c->engine

UTILITY METHODS

$c->path_to(@path)

Merges @path with $c->config->{home} and returns a Path::Class::Dir object. Note you can usually use this object as a filename, but sometimes you will have to explicitly stringify it yourself by calling the ->stringify method.

For example:

$c->path_to( 'db', 'sqlite.db' );

$c->plugin( $name, $class, @args )

Helper method for plugins. It creates a class data accessor/mutator and loads and instantiates the given class.

Note: This method of adding plugins is deprecated. The ability to add plugins like this will be removed in a Catalyst 5.81. Please do not use this functionality in new code.

MyApp->setup

Initializes the dispatcher and engine, loads any plugins, and loads the model, view, and controller components. You may also specify an array of plugins to load here, if you choose to not load them in the use Catalyst line.

MyApp->setup;
MyApp->setup( qw/-Debug/ );

$app->setup_finalize

A hook to attach modifiers to. This method does not do anything except set the setup_finished accessor.

Applying method modifiers to the setup method doesn't work, because of quirky thingsdone for plugin setup.

Example:

after setup_finalize => sub {
my $app = shift;
## do stuff here..
};

$c->uri_for( $path?, @args?, \%query_values? )

$c->uri_for( $action, \@captures?, @args?, \%query_values? )

Constructs an absolute URI object based on the application root, the provided path, and the additional arguments and query parameters provided. When used as a string, provides a textual URI.

If no arguments are provided, the URI for the current action is returned. To return the current action and also provide @args, use $c->uri_for( $c->action, @args ).

If the first argument is a string, it is taken as a public URI path relative to $c->namespace (if it doesn't begin with a forward slash) or relative to the application root (if it does). It is then merged with $c->request->base; any @args are appended as additional path components; and any %query_values are appended as ?foo=bar parameters.

If the first argument is a Catalyst::Action it represents an action which will have its path resolved using $c->dispatcher->uri_for_action. The optional \@captures argument (an arrayref) allows passing the captured variables that are needed to fill in the paths of Chained and Regex actions; once the path is resolved, uri_for continues as though a path was provided, appending any arguments or parameters and creating an absolute URI.

The captures for the current request can be found in $c->request->captures, and actions can be resolved using Catalyst::Controller->action_for($name). If you have a private action path, use $c->uri_for_action instead.

$c->log_headers($type => $headers)

$c->prepare_read

$c->prepare_request

$c->prepare_uploads

Prepares uploads.

$c->prepare_write

Prepares the output for writing.

$c->request_class

Returns or sets the request class.

$c->response_class

Returns or sets the response class.

$c->read( [$maxlength] )

Reads a chunk of data from the request body. This method is designed to be used in a while loop, reading $maxlength bytes on every call. $maxlength defaults to the size of the request if not specified.

You have to set MyApp->config(parse_on_demand => 1) to use this directly.

Warning: If you use read(), Catalyst will not process the body, so you will not be able to access POST parameters or file uploads via $c->request. You must handle all body parsing yourself.

$c->run

Starts the engine.

$c->set_action( $action, $code, $namespace, $attrs )

Sets an action in a given namespace.

$c->setup_actions($component)

Sets up actions for a component.

$c->setup_components

This method is called internally to set up the application's components.

The setup_components config option is passed to both of the above methods.

Installation of each component is performed by the setup_component method, below.

$c->locate_components( $setup_component_config )

This method is meant to provide a list of component modules that should be setup for the application. By default, it will use Module::Pluggable.

Specify a setup_components config option to pass additional options directly to Module::Pluggable. To add additional search paths, specify a key named search_extra as an array reference. Items in the array beginning with :: will have the application class name prepended to them.

$c->expand_component_module( $component, $setup_component_config )

Components found by locate_components will be passed to this method, which is expected to return a list of component (package) names to be set up.

$c->setup_component

$c->setup_dispatcher

Sets up dispatcher.

$c->setup_engine

Sets up engine.

$c->setup_home

Sets up the home directory.

$c->setup_log

Sets up log by instantiating a Catalyst::Log object and passing it to log(). Pass in a comma-delimited list of levels to set the log to.

This method also installs a debug method that returns a true value into the catalyst subclass if the "debug" level is passed in the comma-delimited list, or if the $CATALYST_DEBUG environment variable is set to a true value.

Note that if the log has already been setup, by either a previous call to setup_log or by a call such as __PACKAGE__->log( MyLogger->new ), that this method won't actually set up the log object.

$c->setup_plugins

Sets up plugins.

$c->setup_stats

Sets up timing statistics class.

$c->registered_plugins

Returns a sorted list of the plugins which have either been stated in the import list or which have been added via MyApp->plugin(@args);.

If passed a given plugin name, it will report a boolean value indicating whether or not that plugin is loaded. A fully qualified name is required if the plugin name does not begin with Catalyst::Plugin::.

if ($c->registered_plugins('Some::Plugin')) {
...
}

$c->stack

Returns an arrayref of the internal execution stack (actions that are currently executing).

$c->stats

Returns the current timing statistics object. By default Catalyst uses Catalyst::Stats, but can be set otherwise with stats_class.

Even if -Stats is not enabled, the stats object is still available. By enabling it with $c-stats->enabled(1) >, it can be used to profile explicitly, although MyApp.pm still won't profile nor output anything by itself.

$c->stats_class

Returns or sets the stats (timing statistics) class. Catalyst::Stats is used by default.

CONFIGURATION

default_view - The default view to be rendered or returned when $c->view is called. See "$c->view($name)".

disable_component_resolution_regex_fallback - Turns off the deprecated component resolution functionality so that if any of the component methods (e.g. $c->controller('Foo')) are called then regex search will not be attempted on string values and instead undef will be returned.

home - The application home directory. In an uninstalled application, this is the top level application directory. In an installed application, this will be the directory containing MyApp.pm.

name - The name of the application in debug messages and the debug and welcome screens

parse_on_demand - The request body (for example file uploads) will not be parsed until it is accessed. This allows you to (for example) check authentication (and reject the upload) before actually recieving all the data. See "ON-DEMAND PARSER"

root - The root directory for templates. Usually this is just a subdirectory of the home directory, but you can set it to change the templates to a different directory.

search_extra - Array reference passed to Module::Pluggable to for additional namespaces from which components will be loaded (and constructed and stored in $c->components).

show_internal_actions - If true, causes internal actions such as _DISPATCH to be shown in hit debug tables in the test server.

use_request_uri_for_path - Controlls if the REQUEST_URI or PATH_INFO environment variable should be used for determining the request path. See "PATH DECODING" in Catalyst::Engine::CGI for more information.

INTERNAL ACTIONS

Catalyst uses internal actions like _DISPATCH, _BEGIN, _AUTO, _ACTION, and _END. These are by default not shown in the private action table, but you can make them visible with a config parameter.

MyApp->config(show_internal_actions => 1);

ON-DEMAND PARSER

The request body is usually parsed at the beginning of a request, but if you want to handle input yourself, you can enable on-demand parsing with a config parameter.

MyApp->config(parse_on_demand => 1);

PROXY SUPPORT

Many production servers operate using the common double-server approach, with a lightweight frontend web server passing requests to a larger backend server. An application running on the backend server must deal with two problems: the remote user always appears to be 127.0.0.1 and the server's hostname will appear to be localhost regardless of the virtual host that the user connected through.

Catalyst will automatically detect this situation when you are running the frontend and backend servers on the same machine. The following changes are made to the request.

$c->req->address is set to the user's real IP address, as read from
the HTTP X-Forwarded-For header.
The host value for $c->req->base and $c->req->uri is set to the real
host, as read from the HTTP X-Forwarded-Host header.

Additionally, you may be running your backend application on an insecure connection (port 80) while your frontend proxy is running under SSL. If there is a discrepancy in the ports, use the HTTP header X-Forwarded-Port to tell Catalyst what port the frontend listens on. This will allow all URIs to be created properly.

In the case of passing in:

X-Forwarded-Port: 443

All calls to uri_for will result in an https link, as is expected.

Obviously, your web server must support these headers for this to work.

In a more complex server farm environment where you may have your frontend proxy server(s) on different machines, you will need to set a configuration option to tell Catalyst to read the proxied data from the headers.

MyApp->config(using_frontend_proxy => 1);

If you do not wish to use the proxy support at all, you may set:

MyApp->config(ignore_frontend_proxy => 1);

THREAD SAFETY

Catalyst has been tested under Apache 2's threading mpm_worker, mpm_winnt, and the standalone forking HTTP server on Windows. We believe the Catalyst core to be thread-safe.

If you plan to operate in a threaded environment, remember that all other modules you are using must also be thread-safe. Some modules, most notably DBD::SQLite, are not thread-safe.